NAG Fortran Library Routine Document

G05RAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05RAF sets up a reference vector and generates an array of pseudo-random numbers from a Normal (Gaussian) copula with covariance matrix C.

2 Specification

```
SUBROUTINE G05RAF (MODE, M, C, LDC, N, X, LDX, IGEN, ISEED, R, LR,
1                  IFAIL)
INTEGER            MODE, M, LDC, N, LDX, IGEN, ISEED(4), LR, IFAIL
double precision   C(LDC,M), X(LDX,M), R(LR)
```

3 Description

The Gaussian copula, c, is defined by

$$c(u_1, u_2, \cdots, u_m; C) = \Phi_C\left(\phi_{C_{11}}^{-1}(u_1), \phi_{C_{22}}^{-1}(u_2), \cdots, \phi_{C_{mm}}^{-1}(u_m)\right)$$

where m is the number of dimensions, Φ_C is the multivariate Normal density function with mean zero and covariance matrix C and $\phi_{C_{ii}}^{-1}$ is the inverse of the univariate Normal density function with mean zero and variance C_{ii}.

Routine G05LYF is used to generate a vector from a multivariate Normal distribution and routine G01EAF is used to convert each element of that vector into a uniformly distributed value between zero and one.

One of the initialization routines G05KBF (for a repeatable sequence if computed sequentially) or G05KCF (for a non-repeatable sequence) must be called prior to the first call to G05RAF.

4 References

Nelsen R B (1998) *An Introduction to Copulas. Lecture Notes in Statistics 139* Springer

Sklar A (1973) Random Variables: Joint Distribution Functions and Copulas *Kybernetika* **9** 499–460

5 Parameters

1: MODE – INTEGER *Input*

On entry: selects the operation to be performed:

MODE $= 0$

Initialize and generate random numbers.

MODE $= 1$

Initialize only (i.e., set up reference vector).

MODE $= 2$

Generate random numbers using previously set up reference vector.

Constraint: $0 \leq \text{MODE} \leq 2$.

2: M – INTEGER *Input*

On entry: m, the number of dimensions of the distribution.

Constraint: $\mathrm{M} > 0$.

3: C(LDC,M) – ***double precision*** array *Input*

On entry: the covariance matrix of the distribution. Only the upper triangle need be set.

Constraint: C must be positive semi-definite to ***machine precision***.

4: LDC – INTEGER *Input*

On entry: the first dimension of the array C as declared in the (sub)program from which G05RAF is called.

Constraint: $\mathrm{LDC} \geq \mathrm{M}$.

5: N – INTEGER *Input*

On entry: n, the number of random variates required.

Constraint: $\mathrm{N} \geq 1$.

6: X(LDX,M) – ***double precision*** array *Output*

On exit: the array of pseudo-random multivariate Normal vectors generated by the routine.

7: LDX – INTEGER *Input*

On entry: the first dimension of the array X as declared in the (sub)program from which G05RAF is called.

Constraint: $\mathrm{LDX} \geq \mathrm{N}$.

8: IGEN – INTEGER *Input*

On entry: must contain the identification number for the generator to be used to return a pseudo-random number and should remain unchanged following initialization by a prior call to one of the routines G05KBF or G05KCF.

9: ISEED(4) – INTEGER array *Input/Output*

On entry: contains values which define the current state of the selected generator.

On exit: contains updated values defining the new state of the selected generator.

10: R(LR) – ***double precision*** array *Input/Output*

On entry: if $\mathrm{MODE} = 2$, the reference vector as set up by G05RAF in a previous call with MODE $= 0$ or 1.

On exit: if MODE $= 0$ or 1, the reference vector that can be used in subsequent calls to G05RAF with $\mathrm{MODE} = 2$.

11: LR – INTEGER *Input*

On entry: the dimension of the array R as declared in the (sub)program from which G05RAF is called. If $\mathrm{MODE} = 2$, it must be the same as the value of LR specified in the prior call to G05RAF with MODE $= 0$ or 1.

Constraint: $\mathrm{LR} > \mathrm{M}(\mathrm{M}+1)$.

12: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, MODE $\neq 0$, 1 or 2.

IFAIL $= 2$

On entry, M < 1.

IFAIL $= 3$

The covariance matrix C is not positive semi-definite to machine precision.

IFAIL $= 4$

On entry, LDC $<$ M.

IFAIL $= 5$

On entry, N < 1.

IFAIL $= 7$

On entry, LDX $<$ N.

IFAIL $= 8$

On entry, invalid value for IGEN. IGEN must be the same as the value as specified in the prior call to G05RAF with MODE $= 0$ or 1.

IFAIL $= 10$

The reference vector R has been corrupted or M has changed since R was set up in a previous call with MODE $= 0$ or 1.

IFAIL $= 11$

On entry, LR $<$ M$(\mathrm{M}+1)$.

7 Accuracy

See Section 7 of the document for G05LYF for an indication of the accuracy of the underlying multivariate Normal distribution.

8 Further Comments

None.

9 Example

The example program prints ten pseudo-random observations from a Normal copula with covariance matrix

$$\begin{bmatrix} 1.69 & 0.39 & -1.86 & 0.07 \\ 0.39 & 98.01 & -7.07 & -0.71 \\ -1.86 & -7.07 & 11.56 & 0.03 \\ 0.07 & -0.71 & 0.03 & 0.01 \end{bmatrix},$$

generated by G05RAF. All ten observations are generated by a single call to G05RAF with MODE = 0. The random number generator is initialized by G05KBF.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     G05RAF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NOUT, LDC, LDX, LR
      PARAMETER        (NOUT=6,LDC=5,LDX=100,LR=LDC*LDC+LDC+1)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, IGEN, J, M, N
*     .. Local Arrays ..
      DOUBLE PRECISION C(LDC,LDC), R(LR), X(LDX,LDC)
      INTEGER          ISEED(4)
*     .. External Subroutines ..
      EXTERNAL         G05KBF, G05RAF
*     .. Executable Statements ..
      CONTINUE

      WRITE (NOUT,*) 'G05RAF Example Program Results'
      WRITE (NOUT,*)

*     Initialise the seed to a repeatable sequence
      ISEED(1) = 1762543
      ISEED(2) = 9324783
      ISEED(3) = 42344
      ISEED(4) = 742355

*     Choose the random generator to use
      IGEN = 1

*     Initialise the random generator
      CALL G05KBF(IGEN,ISEED)

*     Set the number of variables and variates
      M = 4
      N = 10

*     Input the upper triangle portion of the covariance matrix
      C(1,1) = 1.69D0
      C(1,2) = 0.39D0
      C(1,3) = -1.86D0
      C(1,4) = 0.07D0
      C(2,2) = 98.01D0
      C(2,3) = -7.07D0
      C(2,4) = -0.71D0
      C(3,3) = 11.56D0
      C(3,4) = 0.03D0
      C(4,4) = 0.01D0

      IFAIL = 0

*     Set up reference vector and generate N numbers
      CALL G05RAF(0,M,C,LDC,N,X,LDX,IGEN,ISEED,R,LR,IFAIL)
```

```
*     Display the results
      DO 20 I = 1, N
         WRITE (NOUT,99999) (X(I,J),J=1,M)
   20 CONTINUE

      STOP

99999 FORMAT (1X,10F10.4)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 G05RAF Example Program Results

     0.9819    0.1689    0.0712    0.9428
     0.2525    0.7025    0.5261    0.1134
     0.4771    0.8504    0.8684    0.4064
     0.0900    0.4690    0.7162    0.2592
     0.3835    0.0400    0.8975    0.7915
     0.5492    0.9685    0.4275    0.1749
     0.2064    0.3430    0.9595    0.5140
     0.7369    0.6728    0.0191    0.3639
     0.8970    0.0732    0.5617    0.9151
     0.2198    0.0157    0.8901    0.8911
```